Shuttlecocks and the Akaike Information Criterion

Kurt Bryan and SIMIODE


```
> restart;
  with(plots) :
```

A worksheet to help explore the project in Section 3.5.4.

First, the data for the shuttlecock's fall, in time (seconds)/distance (meters) pairs:

```
> shuttledata := [[0, 0], [0.347, 0.61], [0.47, 1.00], [0.519, 1.22], [0.582, 1.52], [0.650, 1.83],
      [0.674, 2.00], [0.717, 2.13], [0.766, 2.44], [0.823, 2.74], [0.870, 3.00], [1.031, 4.00],
      [1.193, 5.00], [1.354, 6.00], [1.501, 7.00], [1.726, 8.50], [1.873, 9.50]] :
```

Number of data points

```
> n := nops(shuttledata)
```

A plot

```
> plt1 := pointplot(shuttledata, symbol = solidcircle, color = red, symbolsize = 20, labels
      = ["time (seconds)", "distance (meters)"], labeldirections = [horizontal, vertical])
```

We might posit a model of the form v'(t) = g (no air resistance) and consider g as an unknown, to be estimated. Then the governing ODE is (from equation (3.68) in the text)

```
> de := v'(t) = g
```

The solution with v(0) = 0 is

```
> sol := rhs(dsolve({de, v(0) = 0}, v(t)))
```

Make this into a function of t

```
> vsol := unapply(sol, t);
```

The distance fallen x(t) by the shuttlecock is (using x(0) = 0)

```
> xx := int(vsol(tau), tau = 0 ..t) : #Integrate for position
```

Make this into a function of t

```
> x := unapply(xx, t);
```

Form a sum of squares

```
> SS := add((x(shuttledata[j, 1]) - shuttledata[j, 2])^2, j = 1 ..n)
```

Minmize in g. First, a plot

```
> plot(SS, g = 0 ..15)
```

Solve SS'(g) = 0 to find the least-squares estimate for gravitational acceleration

```
> eq := diff(SS, g) = 0;
  gest := solve(eq, g)
```

The residual is

```
> subs(g = gest, SS)
```

A plot to compare the fit of this model to the data:

```
> plt2 := plot(subs(g = gest, x(t)), t = 0 ..1.873, color = blue) :
> display(plt1, plt2)
>
```